

Hwk #7 Solution Notebook

Morin 5.9

If a particle moves in one dimension under the influence of a force $F(x) = -V'(x)$, then at any place where the force drops to zero (i.e. any x_0 such that $F(x_0)=0$) it is a possible "motion" for the particle to stay put (i.e. $x(t) = x_0$). If the particle is displaced a small amount from x_0 , though, what happens?

Expanding $F(x)$ in a Taylor series near x_0 : $F(x) = F(x_0) + F'(x_0)(x-x_0) + \dots = 0 + F'(x_0)(x-x_0) + \dots$

We obtain the force law of a spring with equilibrium point x_0 and spring constant $-F'(x_0)$.

So a very common question to be asked is: given $V(x)$ where are its minima (i.e. find the x_0 such that $V'(x_0)=0$), and for each minimum what is the frequency of small oscillations about that minimum (i.e. what is $k_{\text{eff}} = V''(x_0)$, and then $\omega = \text{sqrt}[k_{\text{eff}}/m]$).

Here we are asked these questions for

```
In[1]:= V[x_] = - C x^n Exp[-alpha x]
```


```
Out[1]= - C e^{-x alpha} x^n
```

First let's plot this thing:

```
In[2]:= p = {C -> 1, n -> 3, alpha -> 1}
```

```
Out[2]= {C -> 1, n -> 3, alpha -> 1}
```

```
In[3]:= plotV = Plot[V[x] /. p, {x, 0, 10}]
```



Next locate the minimum:

```
In[4]:= eqn0 = V'[x] == 0
```

```
Out[4]= - C e^{-x alpha} n x^{-1+n} + C e^{-x alpha} x^n alpha == 0
```

```
In[5]:= soln0 = Solve[eqn0, x]
```

```
Out[5]= {{x -> n/alpha}}
```

And compute the 2nd derivative there:

```
In[6]:= V''[x] /. soln0[[1]]
```

$$\text{Out[6]} = -C e^{-n} (-1+n) n \left(\frac{n}{\alpha}\right)^{-2+n} + 2 C e^{-n} n \left(\frac{n}{\alpha}\right)^{-1+n} \alpha - C e^{-n} \left(\frac{n}{\alpha}\right)^n \alpha^2$$

```
In[7]:= keff = Simplify[%]
```

$$\text{Out[7]} = C e^{-n} \left(\frac{n}{\alpha}\right)^{-1+n} \alpha$$

So to answer the question posed, the (angular) frequency of small oscillations is

```
In[8]:= ω0 = Sqrt[keff / m]
```

$$\text{Out[8]} = \sqrt{\frac{C e^{-n} \left(\frac{n}{\alpha}\right)^{-1+n} \alpha}{m}}$$

And the period would then be

```
In[9]:= T0 = 2 π / ω0
```

$$\text{Out[9]} = \frac{2 \pi}{\sqrt{\frac{C e^{-n} \left(\frac{n}{\alpha}\right)^{-1+n} \alpha}{m}}}$$

So let's choose parameters and actually find the motion. First attempt:

```
In[10]:= p = {C → 1, n → 3, α → 1, m → 1}
```

```
Out[10]= {C → 1, n → 3, α → 1, m → 1}
```

```
In[11]:= eqnFma = m x''[t] == -V'[x[t]]
```

```
Out[11]= m x''[t] == C e^{-α x[t]} n x[t]^{-1+n} - C e^{-α x[t]} α x[t]^n
```

```
In[12]:= eqnFma /. p
```

```
Out[12]= x''[t] == 3 e^{-x[t]} x[t]^2 - e^{-x[t]} x[t]^3
```

Wait as long as you like for this one to come back, then hit Alt-plus to abort the calculation:

```
DSolve[{eqnFma /. p}, x[t], t]
```

```
$Aborted
```

Even when the differential equation is too hard for *Mathematica*, we can get a numerical solution.

Let's launch the motion from the equilibrium point with some choice of initial velocity:

```
In[13]:= x0 = x /. soln0[[1]] /. p
```

```
Out[13]= 3
```


```
In[14]:= sN = NDSolve[{eqnFma /. p, x[0] == x0, x'[0] == .1}, x[t], {t, 0, 2 T0 /. p}]
```

```
Out[14]= {{x[t] → InterpolatingFunction[{{0., 18.7729}}, <>][t]}}
```

```
In[15]:= x1[t_] = x[t] /. sN[[1]]
```

```
Out[15]= InterpolatingFunction[{{0., 18.7729}}, <>][t]
```

```
In[16]:= Plot[x1[t], {t, 0, 2 T0 /. p}]
```



So for a "small" velocity (.1) this looks pretty much like a sine.
Let's find the exact period and compare

```
In[17]:= FindRoot[x1[t] == x0, {t, 9}]
```

Out[17]= {t → 9.40053}

```
In[18]:= T0 /. p // N
```

Out[18]= 9.38643

For larger energies, the motion will not look so much like a sine:


```
In[19]:= sN2 = NDSolve[{eqnFma /. p, x[0] == x0, x'[0] == 1}, x[t], {t, 0, 2 T0 /. p}]
```

Out[19]= {{x[t] → InterpolatingFunction[{{0., 18.7729}}, <>][t]}}

```
In[20]:= x2[t_] = x[t] /. sN2[[1]]
```

Out[20]= InterpolatingFunction[{{0., 18.7729}}, <>][t]

```
In[21]:= Plot[x2[t], {t, 0, 2 T0 /. p}]
```




Let's animate the 1D motion on top of the graph of the potential.

For maximum artistry, let's choose to draw this motion at the appropriate energy level:


```
In[22]:= E0 = (V[x0] + (1 / 2) m x2'[0]) /. p
```

Out[22]= -0.844251

```
In[23]:= plotE0 = Plot[E0, {x, 1, 9}]
```



```
In[24]:= Animate[Show[plotV, plotE0, ListPlot[{{x2[t], E0}}, PlotStyle -> {Red, PointSize[.02]}], {t, 0, 2 T0 / . p}]
```



Morin 5.15 and 5.16

Between time t and time $t+\Delta t$ a snowball of mass $\Delta m = \sigma \Delta t$ moving at velocity u collides with a car of mass M moving at $v(t)$. In the car frame the speed of the incoming snowball is $u-v$. Assuming the collision is elastic (not so realistic for a snowball), the outgoing speed is $(v-u)$ in the car frame, and therefore $2v-u$ in the ground frame. Conservation of momentum then reads:

```
In[25]:= Δp = (M (v + Δv) + Δm (2 v - u)) - (M v + Δm u)
```

```
Out[25]= -M v - u Δm + (-u + 2 v) Δm + M (v + Δv)
```

```
In[26]:= s = Solve[Δp == 0, Δv] [[1]]
```

```
Out[26]= {Δv -> (2 (u Δm - v Δm)) / M}
```

Next divide by Δt to get the equation of motion:

In[27]:= $\Delta v / \Delta t /. s$

$$\text{Out[27]= } \frac{2 (u \Delta m - v \Delta m)}{M \Delta t}$$

In[28]:= $\text{eqn} = v'[t] == (\Delta v / \Delta t /. s /. \{\Delta m \rightarrow (\sigma (u - v[t]) / u) \Delta t, v \rightarrow v[t]\}) // \text{Simplify}$

$$\text{Out[28]= } v'[t] == \frac{2 \sigma (u - v[t])^2}{M u}$$

And now we can ask for the general solution:

In[29]:= $\text{DSolve}[\text{eqn}, v[t], t]$

$$\text{Out[29]= } \left\{ \left\{ v[t] \rightarrow \frac{-M u + 2 t u \sigma + M u^2 C[1]}{2 t \sigma + M u C[1]} \right\} \right\}$$

And specifically if we start from rest at time $t=0$ here is the velocity:

In[30]:= $\text{sv1} = \text{DSolve}[\{\text{eqn}, v[0] == 0\}, v[t], t][[1]]$

$$\text{Out[30]= } \left\{ v[t] \rightarrow \frac{2 t u \sigma}{M + 2 t \sigma} \right\}$$

And position:

In[31]:= $\text{x1}[\text{tf}_] = \text{Integrate}[v[t] /. \text{sv1}, \{t, 0, \text{tf}\}, \text{Assumptions} \rightarrow \{M > 0, \text{tf} > 0, \sigma > 0\}]$


$$\text{Out[31]= } \text{tf} u + \frac{M u \text{Log}\left[\frac{M}{M+2 \text{tf} \sigma}\right]}{2 \sigma}$$

Choosing numerical parameters we can plot:


In[32]:= $p = \{u \rightarrow 1, M \rightarrow 1, \sigma \rightarrow 1\}$

Out[32]= $\{u \rightarrow 1, M \rightarrow 1, \sigma \rightarrow 1\}$

In[33]:= $\text{Plot}[v[t] /. \text{sv1} /. p, \{t, 0, 3\}]$



In[34]:= `Plot[x1[t] /. p, {t, 0, 3}]`



In problem 5.16 we open the window and let the snowballs accumulate.

In[35]:= `$\Delta p = (m (v + \Delta v) + \Delta m (v)) - (m v + \Delta m u)$`

Out[35]= $-m v - u \Delta m + v \Delta m + m (v + \Delta v)$

In[36]:= `$s = \text{Solve}[\Delta p == 0, \Delta v] [[1]]$`

Out[36]= $\left\{ \Delta v \rightarrow \frac{u \Delta m - v \Delta m}{m} \right\}$

In[37]:= `$\text{eqnv} = v'[t] == (\Delta v / \Delta t /. s /. \{\Delta m \rightarrow (\sigma (u - v[t]) / u) \Delta t, v \rightarrow v[t], m \rightarrow m[t]\} // \text{Simplify})$`

Out[37]= $v'[t] == \frac{\sigma (u - v[t])^2}{u m[t]}$

Now the mass is a function of time, which obeys the differential equation:

In[38]:= `$\text{eqnm} = m'[t] == \sigma (u - v[t]) / u$`

Out[38]= $m'[t] == \frac{\sigma (u - v[t])}{u}$

Solving for $M[t]$ and $v[t]$ together:

In[39]:= `$\text{soln} = \text{Simplify}[\text{DSolve}[\{\text{eqnv}, \text{eqnm}, m[0] == M, v[0] == 0\}, \{v[t], m[t]\}, t], \text{Assumptions} \rightarrow u > 0]$`

Out[39]= $\left\{ \left\{ m[t] \rightarrow -\sqrt{M (M + 2 t \sigma)}, v[t] \rightarrow \frac{u \left(M + 2 t \sigma + \sqrt{M (M + 2 t \sigma)} \right)}{M + 2 t \sigma} \right\}, \left\{ m[t] \rightarrow \sqrt{M (M + 2 t \sigma)}, v[t] \rightarrow \frac{u \left(M + 2 t \sigma - \sqrt{M (M + 2 t \sigma)} \right)}{M + 2 t \sigma} \right\} \right\}$

Plainly we are interested in the solution with positive mass. To get the position,

In[40]:= `$\text{sv2} = \text{soln} [[2]]$`

Out[40]= $\left\{ m[t] \rightarrow \sqrt{M (M + 2 t \sigma)}, v[t] \rightarrow \frac{u \left(M + 2 t \sigma - \sqrt{M (M + 2 t \sigma)} \right)}{M + 2 t \sigma} \right\}$

```
In[41]:= pos = Integrate[v[t] /. sv2, t]
```

$$\text{Out[41]} = u \left(t - \frac{\sqrt{M(M + 2 t \sigma)}}{\sigma} \right)$$


Starting from $x=0$ at $t=0$:

```
In[42]:= x2[t_] = pos - (pos /. t -> 0)
```

$$\text{Out[42]} = \frac{\sqrt{M^2} u}{\sigma} + u \left(t - \frac{\sqrt{M(M + 2 t \sigma)}}{\sigma} \right)$$


And finally plotting both 5.15 and 5.16 together, we see that for the same parameters, the superballs make a better propellant:

```
In[43]:= Plot[Evaluate[{v[t] /. sv1, v[t] /. sv2} /. p],
  {t, 0, 3}, PlotStyle -> {Red, Blue}, PlotRange -> {0, 1}]
```



Let's take a look at the velocity *difference*, and note it has a max a some finite time:

```
In[44]:= Plot[Evaluate[{(v[t] /. sv1) - (v[t] /. sv2)} /. p], {t, 0, 10}]
```



Falling Chain (Morin 5.28)

A chain of length L and mass density σ is held vertically with the bottom end attached to a support. The upper end is then released and the chain falls. First let's animate the motion. Each link is assumed just to fall at g until it reaches its resting place, so at time t a length $(1/4) g t^2$ is at rest:

Here R is a parameter to control the size of the semicircular loop. A fancy version would stop drawing this bit at the end of the motion, but we aren't fancy.

```
In[46]:= p = {L → 1, g → 1, σ → 1, R → .02}
```

```
Out[46]= {L → 1, g → 1, σ → 1, R → 0.02}
```

```
In[47]:= tf = t /. Solve[(1/4) g t^2 == L, t][[2]]
```

```
Out[47]=  $\frac{2\sqrt{L}}{\sqrt{g}}$ 
```


Here $y(t)$ is the length which is currently dangling:

```
In[48]:= y[t_] = (1/4) g t^2
```

```
Out[48]=  $\frac{g t^2}{4}$ 
```

```
In[49]:= draw[t_] := Show[ParametricPlot[(R {Cos[α], Sin[α]} + {0, -y[t]}) /. p,
  {α, 0, -π}, Axes → False, PlotRange → Evaluate[{-L, L} /. p], Graphics[
  {Line[{R, 0}, {R, -y[t]}) /. p, Line[Evaluate[{-R, L - 2 y[t]}, {-R, -y[t]}) /. p]}]]]
```

In[50]:= `Animate[draw[t], {t, 0, tf /. p}]`



Out[50]=

The homework question is to find the force on the chain applied at the fixed support. Obviously part of this is the force needed to hold up a chain of length $y(t)$, namely $\sigma y g$. The other part is the impulsive force needed to stop the next little bit of chain. Between time t and $t+\Delta t$ a length

In[51]:= `$\Delta y = y'[t] \Delta t$`

Out[51]=
$$\frac{g t \Delta t}{2}$$

comes to rest. Since it was moving at speed $g*t$, the momentum change is:

In[52]:= `$\Delta p = \sigma \Delta y g t$`

Out[52]=
$$\frac{1}{2} g^2 t^2 \Delta t \sigma$$

And the impulsive force is

In[53]:= `$F_{imp}[t_] = \Delta p / \Delta t$`


Out[53]=
$$\frac{1}{2} g^2 t^2 \sigma$$

which makes the total force:

```
In[54]:= F[t_] = Piecewise[{{Fimp[t] + σ y[t] g, t <= tf}, {σ L g, t > tf}}]
```

$$\text{Out[54]= } \begin{cases} \frac{3}{4} g^2 t^2 \sigma & t \leq \frac{2\sqrt{L}}{\sqrt{g}} \\ g L \sigma & t > \frac{2\sqrt{L}}{\sqrt{g}} \\ 0 & \text{True} \end{cases}$$

```
In[55]:= Plot[F[t] /. p, {t, 0, 1 + tf /. p}]
```



Speedy Travel (Morin 5.63)

Take an idealized spherical planet of uniform mass density ρ . Drill a straight tube through and make it frictionless. Drop a mass down the tube. Describe the motion, and find out how long you have to wait for the mass to come back.

Soln: Let's use $\{x,y\}$ coordinate system and orient our tube at constant $y(t) = y_0$. Our goal is then to find $x(t)$, starting from rest at $x_0 = \sqrt{R^2 - y_0^2}$. Note that the problem statement didn't specify y_0 or R ---they had better not matter in the end, or the problem was not well-posed.

We'll find it useful along the way to introduce these two functions of x : $r = \sqrt{x^2 + y_0^2}$ and $\theta = \text{ArcSin}[x/r]$

There are two forces acting on our mass: gravity and a normal force which enforces the constraint $y=y_0$.

The y -component of $F=ma$ tells us what N needs to be, while the x -component reads:

$$m x''[t] = - F_g \sin\theta$$

Here F_g is the magnitude of the gravity force when the mass is at location x , i.e. at radius r . To nail that down, recall Gauss's law for gravity, in particular that for a hollow spherical shell of radius R the gravitational field on the exterior is the same as that from a point mass, while the field inside vanishes. Building our uniform sphere out of a collection of hollow spheres, we see that the gravity force at radius r is that due to a point mass of size

```
In[56]:= M[r_] = ρ (4 / 3) π r^3
```

$$\text{Out[56]= } \frac{4}{3} \pi r^3 \rho$$

```
In[57]:= Fg[r_] = G M[r] m / r^2
```

$$\text{Out[57]= } \frac{4}{3} G m \pi r \rho$$

Since $\sin(\theta) = x/r$, our equation of motion is:

$$\text{In[58]:= eqn} = \mathbf{m} \mathbf{x}''[\mathbf{t}] == -\mathbf{Fg}[\mathbf{r}] \mathbf{x}[\mathbf{t}] / \mathbf{r}$$

$$\text{Out[58]:= } m x''[t] == -\frac{4}{3} G m \pi \rho x[t]$$

which we recognize as good old simple harmonic motion:

$$\text{In[59]:= DSolve[eqn, x[t], t]$$

$$\text{Out[59]:= } \left\{ \left\{ x[t] \rightarrow C[1] \cos\left[2\sqrt{G}\sqrt{\frac{\pi}{3}}t\sqrt{\rho}\right] + C[2] \sin\left[2\sqrt{G}\sqrt{\frac{\pi}{3}}t\sqrt{\rho}\right] \right\} \right\}$$

with frequency

$$\text{In[60]:= } \omega_0 = \mathbf{Sqrt}[(4/3)\pi G \rho]$$

$$\text{Out[60]:= } 2\sqrt{\frac{\pi}{3}}\sqrt{G\rho}$$

and therefore period

$$\text{In[61]:= } T = 2\pi / \omega_0$$

$$\text{Out[61]:= } \frac{\sqrt{3}\pi}{\sqrt{G\rho}}$$

Note that indeed the period doesn't depend on the details such as the planet's radius, or even where we drill the tube. Numerically,

$$\text{In[62]:= Needs["PhysicalConstants`"]}$$

$$\text{In[63]:= } \rho_0 = \mathbf{EarthMass} / ((4/3)\pi \mathbf{EarthRadius}^3)$$

$$\text{Out[63]:= } \frac{5496.79 \text{ Kilogram}}{\text{Meter}^3}$$

$$\text{In[64]:= } T /. \{\mathbf{G} \rightarrow \mathbf{GravitationalConstant}, \rho \rightarrow \rho_0\} /. \{\mathbf{Newton} \rightarrow \mathbf{Kilogram Meter / Second}^2\}$$

$$\text{Out[64]:= } \frac{5068.49}{\sqrt{\frac{1}{\text{Second}^2}}}$$

$$\text{In[65]:= Convert[5068.49 Second, Hour]}$$

$$\text{Out[65]:= } 1.40791 \text{ Hour}$$

which is indeed a speedy trip to the other side of the world and back.

(We note that the book problem asks for the one-way time, i.e. half of this).

Amusingly, this is the same period one finds for a satellite orbiting just above the atmosphere:

$$\text{In[66]:= eqnFma} = -m \omega^2 R == (-GMm/R^2 /. M \rightarrow (4/3)\pi \rho R^3)$$

$$\text{Out[66]:= } -m R \omega^2 == -\frac{4}{3} G m \pi R \rho$$

```
In[67]:= Solve[eqnFma,  $\omega$ ]
```

$$\text{Out[67]} = \left\{ \left\{ \omega \rightarrow -2 \sqrt{G} \sqrt{\frac{\pi}{3}} \sqrt{\rho} \right\}, \left\{ \omega \rightarrow 2 \sqrt{G} \sqrt{\frac{\pi}{3}} \sqrt{\rho} \right\} \right\}$$

Displaying the motions together. Choose a planet of radius 1 and $y_0=1/2$

```
In[68]:= circle = ParametricPlot[{Cos[ $\alpha$ ], Sin[ $\alpha$ ]}, { $\alpha$ , 0, 2  $\pi$ }];
```

```
In[69]:=  $y_0 = 1 / 2$ 
```

$$\text{Out[69]} = \frac{1}{2}$$


```
In[70]:=  $x_0 = \text{Sqrt}[1 - y_0^2]$ 
```

$$\text{Out[70]} = \frac{\sqrt{3}}{2}$$

```
In[71]:= tube1 = Graphics[Line[{{ $x_0$ ,  $y_0$ }, {- $x_0$ ,  $y_0$ }]}];
```


```
In[72]:= tube2 = Graphics[Line[{{ $x_0$ ,  $y_0$ }, {- $x_0$ , - $y_0$ }]}];
```

```
In[73]:= Show[circle, tube1, tube2]
```



```
In[74]:= pt1[t_] = { $x_0 \text{Cos}[t]$ ,  $y_0$ };
pt2[t_] = { $x_0 \text{Cos}[t]$ ,  $y_0 \text{Cos}[t]$ };
 $\phi = \text{ArcTan}[y_0 / x_0]$ ;
pt3[t_] = { $\text{Cos}[t + \phi]$ ,  $\text{Sin}[t + \phi]$ };
```

```
Animate[Show[circle, tube1, tube2,
  ListPlot[{pt1[t], pt2[t], pt3[t]}, PlotStyle -> {Red, PointSize[.02]}], {t, 0, 2 Pi}]
```



Morin 5.73

A one dimensional collision: a mass $2m$ moves to the right at v and a mass m move to the left at v . They collide elastically. Find the final velocities (a) working in the lab frame and (b) in the CM frame.

Soln: in the lab frame we write the two relations coming from momentum and energy conservation:

$$\text{In[78]:= eqnP} = 2 m v + m (-v) == m v1 + 2 m v2$$

$$\text{Out[78]= } m v == m v1 + 2 m v2$$

$$\text{In[79]:= eqnK} = (1/2) 2 m (v)^2 + (1/2) m (-v)^2 == (1/2) m v1^2 + (1/2) 2 m v2^2$$

$$\text{Out[79]= } \frac{3 m v^2}{2} == \frac{m v1^2}{2} + m v2^2$$

$$\text{In[80]:= Solve[{eqnP, eqnK}, {v1, v2}]$$

$$\text{Out[80]= } \left\{ \{v1 \rightarrow -v, v2 \rightarrow v\}, \left\{ v1 \rightarrow \frac{5v}{3}, v2 \rightarrow -\frac{v}{3} \right\} \right\}$$

As always we get two solutions to our quadratic, where here the first solution corresponds to nothing happening, i.e. the particles sliding right through each other with unchanged velocities. We are interested in the nontrivial 2nd solution.

To work in the center of mass frame we first need to find the velocity of the CM.

Since $P_{tot} = M_{tot} v_{cm}$,

$$\text{In[81]:= } \mathbf{v_{cm}} = (2 m v - m v) / (2 m + m)$$

$$\text{Out[81]= } \frac{v}{3}$$

The initial state CM velocities are then obtained by subtracting off v_{cm}

$$\text{In[82]:= } \mathbf{v2_{cmInit}} = v - v_{cm}$$

$$\text{Out[82]= } \frac{2 v}{3}$$

$$\text{In[83]:= } \mathbf{v1_{cmInit}} = -v - v_{cm}$$

$$\text{Out[83]= } -\frac{4 v}{3}$$

As a cross check, note that in the CM frame the total momentum is zero:

$$\text{In[84]:= } \mathbf{m v1_{cmInit} + 2 m v2_{cmInit}}$$

$$\text{Out[84]= } 0$$

If the collision is elastic, then the final state consists of both masses just reversing their directions.

To show this the long way, though:

$$\text{In[85]:= } \mathbf{eqnP_{cm}} = m v1_{cmInit} + 2 m v2_{cmInit} == m v1_{cmFinal} + 2 m v2_{cmFinal}$$

$$\text{Out[85]= } 0 == m v1_{cmFinal} + 2 m v2_{cmFinal}$$

$$\text{In[86]:= } \mathbf{eqnK_{cm}} = (1/2) m (v1_{cmInit})^2 + (1/2) 2 m (v2_{cmInit})^2 == (1/2) m (v1_{cmFinal})^2 + (1/2) 2 m (v2_{cmFinal})^2$$

$$\text{Out[86]= } \frac{4 m v^2}{3} == \frac{m v1_{cmFinal}^2}{2} + m v2_{cmFinal}^2$$

$$\text{In[87]:= } \mathbf{s = Solve\{eqnP_{cm}, eqnK_{cm}\}, \{v1_{cmFinal}, v2_{cmFinal}\}}$$

$$\text{Out[87]= } \left\{ \left\{ v1_{cmFinal} \rightarrow -\frac{4 v}{3}, v2_{cmFinal} \rightarrow \frac{2 v}{3} \right\}, \left\{ v1_{cmFinal} \rightarrow \frac{4 v}{3}, v2_{cmFinal} \rightarrow -\frac{2 v}{3} \right\} \right\}$$

Which again has the two solutions: pass on through w/o interaction, or bounce off backwards.

Back in the lab frame we then find final velocities:

$$\text{In[88]:= } \mathbf{v1_{labFinal}} = (v1_{cmFinal} /. s[[2]]) + v_{cm}$$

$$\text{Out[88]= } \frac{5 v}{3}$$

$$\text{In[89]:= } \mathbf{v2_{labFinal}} = (v2_{cmFinal} /. s[[2]]) + v_{cm}$$

$$\text{Out[89]= } -\frac{v}{3}$$

which happily enough agrees with the result above.

■ Morin 5.80

A 2D elastic collision: mass $m_2=2m$ collides with $m_1=m$, and after the collision the two head off at angles $+\theta$ and $-\theta$. Find θ .

Soln: write the three conservation law (for p_x , p_y and kinetic energy), then solve for the three unknowns (v_1 , v_2 and θ)

$$\text{In[90]:= eqnpx} = 2 m v_0 == 2 m v_2 \text{Cos}[\theta] + m v_1 \text{Cos}[\theta]$$

$$\text{Out[90]= } 2 m v_0 == m v_1 \text{Cos}[\theta] + 2 m v_2 \text{Cos}[\theta]$$

$$\text{In[91]:= eqnpy} = 0 == 2 m v_2 \text{Sin}[\theta] + m v_1 \text{Sin}[-\theta]$$

$$\text{Out[91]= } 0 == -m v_1 \text{Sin}[\theta] + 2 m v_2 \text{Sin}[\theta]$$

$$\text{In[92]:= eqnK} = (1/2) 2 m v_0^2 == (1/2) 2 m v_2^2 + (1/2) m v_1^2$$

$$\text{Out[92]= } m v_0^2 == \frac{m v_1^2}{2} + m v_2^2$$

$$\text{In[93]:= Solve}\{\{\text{eqnpx}, \text{eqnpy}, \text{eqnK}\}, \{v_1, v_2, \theta\}\}$$

Solve::ifun :

Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. >>

$$\begin{aligned} \text{Out[93]= } & \left\{ \{v_1 \rightarrow 0, v_2 \rightarrow -v_0, \theta \rightarrow -\pi\}, \{v_1 \rightarrow 0, v_2 \rightarrow -v_0, \theta \rightarrow \pi\}, \{v_1 \rightarrow 0, v_2 \rightarrow v_0, \theta \rightarrow 0\}, \right. \\ & \left\{v_1 \rightarrow -\frac{4 v_0}{3}, v_2 \rightarrow -\frac{v_0}{3}, \theta \rightarrow -\pi\right\}, \left\{v_1 \rightarrow -\frac{4 v_0}{3}, v_2 \rightarrow -\frac{v_0}{3}, \theta \rightarrow \pi\right\}, \left\{v_1 \rightarrow \frac{4 v_0}{3}, v_2 \rightarrow \frac{v_0}{3}, \theta \rightarrow 0\right\}, \\ & \left\{v_1 \rightarrow -\frac{2 v_0}{\sqrt{3}}, v_2 \rightarrow -\frac{v_0}{\sqrt{3}}, \theta \rightarrow -\frac{5 \pi}{6}\right\}, \left\{v_1 \rightarrow -\frac{2 v_0}{\sqrt{3}}, v_2 \rightarrow -\frac{v_0}{\sqrt{3}}, \theta \rightarrow \frac{5 \pi}{6}\right\}, \\ & \left. \left\{v_1 \rightarrow \frac{2 v_0}{\sqrt{3}}, v_2 \rightarrow \frac{v_0}{\sqrt{3}}, \theta \rightarrow -\frac{\pi}{6}\right\}, \left\{v_1 \rightarrow \frac{2 v_0}{\sqrt{3}}, v_2 \rightarrow \frac{v_0}{\sqrt{3}}, \theta \rightarrow \frac{\pi}{6}\right\} \right\} \end{aligned}$$

Note how *Mathematica* gives us the same physical solution multiple times with differing values of the angles.

But there really are only four physically distinct solutions here: (1) nothing happens (stated 3 times),

(2) particles continue along the x-axis at appropriate speeds (stated 3 times as well),

(3) mass 2 scatters at $\pi/6$ (stated twice) and (4) mass scatters at $-\pi/6$. These last two are the ones we want.

Now repeat, allowing the m_2 to be more general:

$$\text{In[94]:= eqnpx} = \alpha m v_0 == \alpha m v_2 \text{Cos}[\theta] + m v_1 \text{Cos}[\theta]$$

$$\text{Out[94]= } m v_0 \alpha == m v_1 \text{Cos}[\theta] + m v_2 \alpha \text{Cos}[\theta]$$

$$\text{In[95]:= eqnpy} = 0 == \alpha m v_2 \text{Sin}[\theta] + m v_1 \text{Sin}[-\theta]$$

$$\text{Out[95]= } 0 == -m v_1 \text{Sin}[\theta] + m v_2 \alpha \text{Sin}[\theta]$$

$$\text{In[96]:= eqnK} = (1/2) \alpha m v_0^2 == (1/2) \alpha m v_2^2 + (1/2) m v_1^2$$

$$\text{Out[96]= } \frac{1}{2} m v_0^2 \alpha == \frac{m v_1^2}{2} + \frac{1}{2} m v_2^2 \alpha$$

```
In[97]:= s = Solve[{eqnpx, eqnpy, eqnK}, {v1, v2, θ}]
```

Solve::ifun :

Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. >>

```
Out[97]= {{v1 -> 0, v2 -> -v0, θ -> -π}, {v1 -> 0, v2 -> -v0, θ -> π}, {v1 -> 0, v2 -> v0, θ -> 0},
  {v1 -> -\frac{2 v0 \alpha}{1 + \alpha}, v2 -> \frac{v0 - v0 \alpha}{1 + \alpha}, \theta -> -\pi}, {v1 -> -\frac{2 v0 \alpha}{1 + \alpha}, v2 -> \frac{v0 - v0 \alpha}{1 + \alpha}, \theta -> \pi},
  {v1 -> \frac{2 v0 \alpha}{1 + \alpha}, v2 -> \frac{-v0 + v0 \alpha}{1 + \alpha}, \theta -> 0}, {v1 -> -\frac{v0 \alpha}{\sqrt{1 + \alpha}}, v2 -> -\frac{v0}{\sqrt{1 + \alpha}}, \theta -> -\text{ArcCos}\left[-\frac{\sqrt{1 + \alpha}}{2}\right]},
  {v1 -> -\frac{v0 \alpha}{\sqrt{1 + \alpha}}, v2 -> -\frac{v0}{\sqrt{1 + \alpha}}, \theta -> \text{ArcCos}\left[-\frac{\sqrt{1 + \alpha}}{2}\right]},
  {v1 -> \frac{v0 \alpha}{\sqrt{1 + \alpha}}, v2 -> \frac{v0}{\sqrt{1 + \alpha}}, \theta -> -\text{ArcCos}\left[\frac{\sqrt{1 + \alpha}}{2}\right]},
  {v1 -> \frac{v0 \alpha}{\sqrt{1 + \alpha}}, v2 -> \frac{v0}{\sqrt{1 + \alpha}}, \theta -> \text{ArcCos}\left[\frac{\sqrt{1 + \alpha}}{2}\right]}}
```

And we see that the when α exceeds 3 the argument of the ArcCos gets larger than 1 and we have no real solution.

Let's animate!

```
In[98]:= x1[t_] = Piecewise[{{0, t < 0}, {v1 t Cos[-θ], t > 0}}]
```

```
Out[98]= { 0          t < 0
  t v1 Cos[θ]    t > 0
  0              True
```

```
In[99]:= y1[t_] = Piecewise[{{0, t < 0}, {v1 t Sin[-θ], t > 0}}]
```

```
Out[99]= { 0          t < 0
 -t v1 Sin[θ]   t > 0
  0              True
```

```
In[100]:= x2[t_] = Piecewise[{{v0 t, t < 0}, {v2 t Cos[θ], t > 0}}]
```

```
Out[100]= { t v0      t < 0
  t v2 Cos[θ]    t > 0
  0              True
```


```
In[101]:= y2[t_] = Piecewise[{{0, t < 0}, {v2 t Sin[θ], t > 0}}]
```

```
Out[101]= { 0          t < 0
  t v2 Sin[θ]   t > 0
  0              True
```

```
In[102]:= p = {v0 -> 1, α -> 1}
```

```
Out[102]= {v0 -> 1, α -> 1}
```


```
In[103]:= Animate[ListPlot[{{x1[t], y1[t]}, {x2[t], y2[t]}} /. s[[10]] /. p,
  PlotStyle -> {Red, PointSize[.03]}, PlotRange -> {{-2, 2}, {-2, 2}}, {t, -2, 2}]
```



And finally give the parameter α a knob:


```
In[104]:= Manipulate[Animate[Show[ListPlot[{{x1[t], y1[t]}} /. s[[10]] /. {v0 -> 1, α -> q},
  PlotStyle -> {Red, PointSize[.03]}, PlotRange -> {{-2, 2}, {-2, 2}},
  ListPlot[{{x2[t], y2[t]}} /. s[[10]] /. {v0 -> 1, α -> q}, PlotStyle -> {Blue, PointSize[.04]},
  PlotRange -> {{-2, 2}, {-2, 2}}]], {t, -2, 2}], {q, 0, 3}]
```



t 

```
Show[ListPlot[{{1.26336 v1 Cos[θ], -1.26336 v1 Sin[θ]} /. {r → - $\frac{2^{2/3} G^{1/3} \left(-\frac{\pi}{3}\right)^{1/3} R \rho^{1/3}}{\omega^{2/3}}$ },
PlotStyle → {RGBColor[1, 0, 0], PointSize[0.03]}, PlotRange → {{-2, 2}, {-2, 2}},
ListPlot[{{1.26336 v2 Cos[θ], 1.26336 v2 Sin[θ]} /. {r → - $\frac{2^{2/3} G^{1/3} \left(-\frac{\pi}{3}\right)^{1/3} R \rho^{1/3}}{\omega^{2/3}}$ }, {r → -
PlotStyle → {RGBColor[0, 0, 1], PointSize[0.04]}, PlotRange → {{-2, 2}, {-2, 2}}]]]
```

Out[104]=



■ Whirling mass

A couple of weeks ago we found the motion of a mass at the end of an ever-shortening rope:

$$\text{In[105]:= } \mathbf{r}[\mathbf{t_}] = \mathbf{r0} - \mathbf{v0} \mathbf{t}$$

$$\text{Out[105]= } \mathbf{r0} - \mathbf{t} \mathbf{v0}$$

Since the force was purely in the radial direction the θ component of $\mathbf{F}=\mathbf{ma}$ reads:

$$\text{In[106]:= } \mathbf{eqn}\theta = \mathbf{m} (\mathbf{r}[\mathbf{t}] \omega'[\mathbf{t}] + 2 \mathbf{r}'[\mathbf{t}] \omega[\mathbf{t}]) == 0$$

$$\text{Out[106]= } \mathbf{m} (-2 \mathbf{v0} \omega[\mathbf{t}] + (\mathbf{r0} - \mathbf{t} \mathbf{v0}) \omega'[\mathbf{t}]) == 0$$

which we DSolve, starting with angular speed ω_0 :

$$\text{In[107]:= } \mathbf{s} = \mathbf{DSolve}[\{\mathbf{eqn}\theta, \omega[0] == \omega_0\}, \omega[\mathbf{t}], \mathbf{t}][[1]]$$

$$\text{Out[107]= } \left\{ \omega[\mathbf{t}] \rightarrow \frac{2 \sqrt{\frac{\pi}{3}} \mathbf{r0}^2 \sqrt{\mathbf{G} \rho}}{(\mathbf{r0} - \mathbf{t} \mathbf{v0})^2} \right\}$$

The radial equation tells us what the the tension has to be:

$$\text{In[109]:= } \mathbf{T} = -\mathbf{m} (\mathbf{r}''[\mathbf{t}] - \mathbf{r}[\mathbf{t}] \omega[\mathbf{t}]^2)$$

$$\text{Out[109]= } \mathbf{m} (\mathbf{r0} - \mathbf{t} \mathbf{v0}) \omega[\mathbf{t}]^2$$

In[110]:= **T / . s**

$$\text{Out[110]} = \frac{4 G m \pi r_0^4 \rho}{3 (r_0 - t v_0)^3}$$

Notice that these expressions are all written as functions of time, but we can also think of them as functions of the radius:

In[111]:= **$\omega[r_]$ = $\omega_0 r_0^2 / r^2$**

$$\text{Out[111]} = \frac{2 \sqrt{\frac{\pi}{3}} r_0^2 \sqrt{G \rho}}{r^2}$$

In[113]:= **T = .**

In[114]:= **T[r_] = m r_0^4 ω_0^2 / r^3**

$$\text{Out[114]} = \frac{4 G m \pi r_0^4 \rho}{3 r^3}$$

Now suppose we reel in the mass from r_0 to a smaller r_1 . The kinetic energy changes from

In[115]:= **K0 = (1/2) m ((r'[0])^2 + (r[0] ω [r0])^2)**

$$\text{Out[115]} = \frac{1}{2} m \left(v_0^2 + \frac{4}{3} G \pi r_0^2 \rho \right)$$

to

In[116]:= **K1 = (1/2) m ((v0)^2 + (r1 ω [r1])^2)**

$$\text{Out[116]} = \frac{1}{2} m \left(v_0^2 + \frac{4 G \pi r_0^4 \rho}{3 r_1^2} \right)$$

Notice that the radial velocity is always v_0 in this problem, so that piece of the kinetic energy is unchanging.

Our job is to compare with the work done by the winch reeling in the rope:

In[117]:= **work = Integrate[-T[r], {r, r0, r1}, Assumptions -> {r0 > 0, r1 > 0}]**

$$\text{Out[117]} = -\frac{2}{3} G m \pi r_0^4 \left(\frac{1}{r_0^2} - \frac{1}{r_1^2} \right) \rho$$

Note that relative to the normal positive r direction the force is $-T$, and the work done is *positive*.

Is it the same as the change in kinetic energy?

In[118]:= **assertion = K1 - K0 == work**

$$\text{Out[118]} = -\frac{1}{2} m \left(v_0^2 + \frac{4}{3} G \pi r_0^2 \rho \right) + \frac{1}{2} m \left(v_0^2 + \frac{4 G \pi r_0^4 \rho}{3 r_1^2} \right) == -\frac{2}{3} G m \pi r_0^4 \left(\frac{1}{r_0^2} - \frac{1}{r_1^2} \right) \rho$$

In[119]:= **Simplify[assertion]**

Out[119]= True

■ The Space Elevator (Morin 5.65)

(a) Describing the earth by its radius (R), density (ρ) and angular frequency ($\omega=2\pi/\text{day}$), find the radius of *geosynchronous* orbit, i.e. that magical * r * at which we "park" satellites.

More precisely they orbit with with angular frequency precisely ω , so they are at rest relative to the spinning Earth.

We are asked to express the answer in terms of a dimensionless number $\eta=r/R$.

Well, first let's define the Earth's mass:

$$\text{In[120]:= } \mathbf{M} = (4 / 3) \pi R^3 \rho$$

$$\text{Out[120]= } \frac{4}{3} \pi R^3 \rho$$

in terms of which the graviational force is the familiar $G M m / r^2$. Equating that with $m a$, where the acceleration is of course $a = (-r \omega^2) \hat{r}$,

Using the relation to express r :

$$\text{In[121]:= } \mathbf{eqn} = - G M m / r^2 == - m r \omega^2$$

$$\text{Out[121]= } - \frac{4 G m \pi R^3 \rho}{3 r^2} == - m r \omega^2$$

$$\text{In[122]:= } \mathbf{s} = \mathbf{Solve}[\mathbf{eqn}, \mathbf{r}]$$

$$\text{Out[122]= } \left\{ \left\{ r \rightarrow - \frac{2^{2/3} G^{1/3} \left(-\frac{\pi}{3}\right)^{1/3} R \rho^{1/3}}{\omega^{2/3}} \right\}, \left\{ r \rightarrow \frac{(-2)^{2/3} G^{1/3} \left(\frac{\pi}{3}\right)^{1/3} R \rho^{1/3}}{\omega^{2/3}} \right\}, \left\{ r \rightarrow \frac{2^{2/3} G^{1/3} \left(\frac{\pi}{3}\right)^{1/3} R \rho^{1/3}}{\omega^{2/3}} \right\} \right\}$$

Obviously we want the 3rd of these cube roots (the others are complex).

And then we get our η cubed:

$$\text{In[123]:= } \mathbf{\eta^3} = (\mathbf{r} / \mathbf{R})^3 /. \mathbf{s}[[3]]$$

$$\text{Out[123]= } \frac{4 G \pi \rho}{3 \omega^2}$$

To convert to a number, let's load up all the physical constants:

$$\text{In[124]:= } \mathbf{Needs}["\mathbf{PhysicalConstants}"]$$

This defines things like

$$\text{In[125]:= } \mathbf{GravitationalConstant}$$

$$\text{Out[125]= } \frac{6.67428 \times 10^{-11} \text{ Meter}^2 \text{ Newton}}{\text{Kilogram}^2}$$

and

$$\text{In[126]:= } \mathbf{EarthRadius}$$

$$\text{Out[126]= } 6378140 \text{ Meter}$$

It doesn't define the density, so let's create our own:

```
In[127]:= ρ0 = EarthMass / ((4 / 3) π EarthRadius ^ 3)
```

```
Out[127]= 
$$\frac{5496.79 \text{ Kilogram}}{\text{Meter}^3}$$

```

and then plug them in:

```
In[128]:= params = {G → GravitationalConstant, ω → (2 π / (3600 * 24 Second)), ρ → ρ0}
```

```
Out[128]= {G →  $\frac{6.67428 \times 10^{-11} \text{ Meter}^2 \text{ Newton}}{\text{Kilogram}^2}$ , ω →  $\frac{\pi}{43200 \text{ Second}}$ , ρ →  $\frac{5496.79 \text{ Kilogram}}{\text{Meter}^3}$ }
```

```
In[129]:= η3 /. params
```

```
Out[129]= 
$$\frac{290.583 \text{ Newton Second}^2}{\text{Kilogram Meter}}$$

```

Sometimes you have to help *Mathematica* along:

```
In[130]:= NewtonHelp = {Newton → Kilogram Meter / Second ^ 2}
```

```
Out[130]= {Newton →  $\frac{\text{Kilogram Meter}}{\text{Second}^2}$ }
```

```
In[131]:= η3 /. params /. NewtonHelp
```

```
Out[131]= 290.583
```

And our numerical expression for η is:

```
In[132]:= η1 = (%) ^ (1 / 3)
```

```
Out[132]= 6.62354
```

Next envision a very long rope of uniform mass density (call it λ , though they didn't specify it---hopefully we won't need it to respond to the questions asked).

It extends from radius R to ηR , where our job is to find η . But we'll call our variable, say, α , since *Mathematica* is fussy about primes.

First let's get the tension on the structure applied at the contact with Earth.

Viewed as a system, the rope has mass $m = \lambda (\alpha - 1) R$, center of mass located at $1 + (\alpha - 1)/2$ in units of R :

```
In[133]:= m = λ (α - 1) R
```

```
Out[133]= R (-1 + α) λ
```

```
In[134]:= rcm = R + R (α - 1) / 2
```

```
Out[134]=  $R + \frac{1}{2} R (-1 + \alpha)$ 
```

And the total gravitational force on the system is the sum of the bits. This requires a quick little integral, which provides as good a time as any to introduce the Assumptions option to Integrate---this is where you can inform *Mathematica* of extra information it can assume is true.

Without it:

In[135]:= **Integrate**[**GM** λ / r^2 , {**r**, **R**, α **R**}]

$$\text{Out[135]= } \frac{4}{3} G \pi R^3 \lambda \rho \text{ If} \left[\text{Re}[\alpha] \geq 0 \mid \mid \alpha \notin \text{Reals}, \frac{-1 + \alpha}{R \alpha}, \right. \\ \left. \text{Integrate} \left[\frac{1}{r^2}, \{r, R, R \alpha\}, \text{Assumptions} \rightarrow !(\text{Re}[\alpha] \geq 0 \mid \mid \alpha \notin \text{Reals}) \right] \right]$$

With it:

In[136]:= **Fg** = **Integrate**[**GM** λ / r^2 , {**r**, **R**, α **R**}, **Assumptions** \rightarrow { $\alpha > 0$ }]

$$\text{Out[136]= } \frac{4 G \pi R^2 (-1 + \alpha) \lambda \rho}{3 \alpha}$$

Together, gravity and tension provide the force necessary for circular motion:

In[139]:= **T0** = .

In[140]:= **eqn** = - **m** ω^2 **r** **cm** == -**Fg** - **T0**

$$\text{Out[140]= } -R \left(R + \frac{1}{2} R (-1 + \alpha) \right) (-1 + \alpha) \lambda \omega^2 == -T0 - \frac{4 G \pi R^2 (-1 + \alpha) \lambda \rho}{3 \alpha}$$

So let's solve for the tension:

In[141]:= **solnT0** = (**Solve**[**eqn**, **T0**] [[1]] // **Simplify**)

$$\text{Out[141]= } \left\{ T0 \rightarrow \frac{R^2 \lambda (-8 G \pi (-1 + \alpha) \rho + 3 \alpha (-1 + \alpha^2) \omega^2)}{6 \alpha} \right\}$$

If our elevator rope is to stand up, the tension can never drop below 0 anywhere, including at $r=R$.

We'll have to check this with the explicit $T(r)$ later on, but it seems plausible that if $T0 > 0$ the rest will be OK.

So for what value of α does $T0$ vanish?

Dividing by the various constants, the polynomial whose roots we want is:

In[142]:= **q** = - **8 G** π (- **1** + α) ρ + **3** α (- **1** + α^2) ω^2

$$\text{Out[142]= } -8 G \pi (-1 + \alpha) \rho + 3 \alpha (-1 + \alpha^2) \omega^2$$

Notice this cubic factorizes---the $\alpha=1$ root is the trivial elevator of zero length:

In[143]:= **Factor**[**q**]

$$\text{Out[143]= } -(-1 + \alpha) (8 G \pi \rho - 3 \alpha \omega^2 - 3 \alpha^2 \omega^2)$$

Adding in the roots of the quadratic, the list of all solutions is:

In[145]:= **soln** = **Solve**[**q** == **0**, α]

$$\text{Out[145]= } \left\{ \{\alpha \rightarrow 1\}, \left\{ \alpha \rightarrow \frac{-3 \omega^2 - \sqrt{3} \sqrt{32 G \pi \rho \omega^2 + 3 \omega^4}}{6 \omega^2} \right\}, \left\{ \alpha \rightarrow \frac{-3 \omega^2 + \sqrt{3} \sqrt{32 G \pi \rho \omega^2 + 3 \omega^4}}{6 \omega^2} \right\} \right\}$$

Plainly we are interesting in the positive solution, which numerically is:

In[146]:= `numa = alpha /. solna[[3]] /. params // N`

$$\text{Out[146]= } 3.15149 \times 10^7 \left(1.73205 \sqrt{\frac{8.39046 \times 10^{-17}}{\text{Second}^4} + \frac{1.9505 \times 10^{-13} \text{ Newton}}{\text{Kilogram Meter Second}^2}} - \frac{1.58655 \times 10^{-8}}{\text{Second}^2} \right) \text{Second}^2$$

And again we need to give a push:

In[147]:= `alpha = Simplify[numa /. {Newton -> Kilogram Meter / Second^2}, Assumptions -> {Second > 0}]`

Out[147]= 23.6126

(c) We are asked at what radius the tension is maximized. Well at such a radius, it is clear that $T'(r) = 0$, which means that the force on a tiny little piece

of string at that location is purely gravitational. If that bit is supposed to keep up with the rest of the elevator, it had better be located at the geosynchronous radius.

But to fully answer this question, we might as well go ahead and get the full $T(r)$.

To get it, we derive and solve a simple differential equation:

Consider the small bit of rope between r and $r+\Delta r$. Its mass is

In[148]:= `Delta m = lambda Delta r`

Out[148]= $\Delta r \lambda$

$F = ma$ for that bit reads:

In[149]:= `eqn = T[r + Delta r] - T[r] - GM Delta m / r^2 == -Delta m omega^2 r`

$$\text{Out[149]= } -\frac{4 G \pi R r^0^4 (-1 + \alpha) \lambda \rho}{3 r^3} - \frac{4 G \pi R^3 \Delta r \lambda \rho}{3 r^2} + \frac{4 G \pi R r^0^4 (-1 + \alpha) \lambda \rho}{3 (r + \Delta r)^3} == -r \Delta r \lambda \omega^2$$

Taking Δr to 0, we get the differential equation

In[159]:= `eqnT = T'[r] == -r lambda omega^2 + GM lambda / r^2`

$$\text{Out[159]= } T'[r] == \frac{4 G \pi R^3 \lambda \rho}{3 r^2} - r \lambda \omega^2$$

Solving with the boundary condition that the tension at the Earth end is T_0 :

In[161]:= `solnT = (DSolve[{eqnT, T[R] == (T0 /. solnT0)}, T[r], r][[1]]) // Simplify`

$$\text{Out[161]= } \left\{ T[r] \rightarrow \frac{(r - R \alpha) \lambda (8 G \pi R^2 \rho - 3 r \alpha (r + R \alpha) \omega^2)}{6 r \alpha} \right\}$$

As a check that we are on the right track, let's get the tension at the far end:

In[162]:= `T[r] /. solnT /. r -> alpha R`

Out[162]= 0

In fact we could have used this condition to find $T(r)$:

In[163]:= `solnT = (DSolve[{eqnT, T[α R] == 0}, T[r], r][[1]]) // Simplify`

$$\text{Out[163]= } \left\{ T[r] \rightarrow \frac{(r - R \alpha) \lambda (8 G \pi R^2 \rho - 3 r \alpha (r + R \alpha) \omega^2)}{6 r \alpha} \right\}$$

Now let's put in numbers and plot. For plotting define the dimensionless parameter $x = r/R$:

In[164]:= `q = (T[r] /. solnT /. α → α0 /. params /. NewtonHelp // Simplify)`

$$\text{Out[164]= } \frac{1}{r \text{ Second}^2} 0.00705839 (r - 23.6126 R) (-3.74625 \times 10^{-7} r^2 - 8.84585 \times 10^{-6} r R + 9.22047 \times 10^{-6} R^2) \lambda$$

In[165]:= `F[x_] = q /. {r → x R, λ → 1 Kilogram / Meter} /. R → EarthRadius // Simplify`


$$\text{Out[165]= } - \frac{107570. \text{ Kilogram Meter } (-23.6126 + x) (-1. + x) (24.6126 + x)}{\text{Second}^2 x}$$

This is the force in Newtons. Let's divide out the units:

In[166]:= `f[x_] = F[x] / Newton /. NewtonHelp`

$$\text{Out[166]= } - \frac{107570. (-23.6126 + x) (-1. + x) (24.6126 + x)}{x}$$

In[167]:= `Plot[f[x], {x, 1, α0}]`



Where is that maximum? Is it in fact at ηR ?

In[168]:= `FindRoot[f'[r] == 0, {r, 7}]`

Out[168]= `{r → 6.62354}`

Yup.